
.OPTIONS  GMIN=1e-18    METHOD=trap   srcsteps = 1  gminsteps = 1
*======== ====== ====== ====== ====== ====== ====== ====== ======
VCC       VC     0      DC     10
VAC1      VIN    0      DC     0      SIN(   0      1   1000 )
I1        VE1    0      1u
I2        VE2    0      1u
QN1       VC1    VB1    VE1    NPN1   1.00
QN2       VC1    VB1    VE2    NPN1   4.25
QN3       VC2    0      VE1    NPN1   4.25
QN4       VC2    0      VE2    NPN1   1.00
R1        VC1    VC     52K
R2        VC2    VC     52K
E_GAIN1   OUT    0      VC1    VC2    1
E_GAININ  VB1    0      VIN    0      1m

.control
tran             25u                   5m         0    1u
plot             out
echo             "THD% versus VIN_vpk "
setplot          new
set              NameList =  (        Nx4 Nx4pt25 Nx4pt5 Nx5  )
compose          NxVals      values   4   4.25    4.5    5
compose          VinVals     values   10m 15m 20m 22m 25m 30m 32m 35m 40m  43m 45m 50m 55m 
60m 70m 80m 90m
settype          voltage     VinVals
let              NoOfNx  =   length(NxVals)
let              NoOfVin =   length(VinVals)

begin
unset            interrupt
* ===============Loop_Nx============
let              j         =  1
while            (j       <=  NoOfNx )
let              Nx        =  NxVals[j-1]
alter            QN2  area =  $&Nx 
alter            QN3  area =  $&Nx 
set              thisName  =  $NameList[$&j]
let              $thisName =  0*vector(NoOfVin)



* ===============Loop_Vin==============
let              k             = 1
while            (k           <= NoOfVin )
let              Vin           = VinVals[k-1]
alter            e_gainin gain = $&Vin
tran             25u              5m        0    1u
linearize
set              specwindow=   "blackman"
spec             200    8k      200     v(out)
let              thdsq =mag(out[9])^2 +mag(out[14])^2  +mag(out[19])^2  +mag(out[24])^2
let              thd_percent= 100*sqrt(thdsq)/mag(out[4]) 
echo             "$&unknown.Vin       $&thd_percent"
let              unknown.{$thisName}[unknown.k-1] = thd_percent
repeat           3
destroy
end
if               ($?interrupt)
goto             bail
endif
let              k =          k + 1
endwhile
setscale         VinVals
plot             $NameList loglog title "DUAL_DIFF THD_% vs Vin_pK and Nx"
let              j =          j + 1
endwhile
label            bail
echo            "Done."
end
.endc

*======== ====== ====== ====== ====== ====== ====== ====== ======
.model    NPN1   NPN(   BF=2100 VAF=216 )
.model    PNP1   PNP(   BF=2100  VAF=21)
.end

=====================END_OF_SPICE=======================

